
10 Anleitung für lokale Sicherung

10.1 DAtenSiCheRung mit ROBOCOPY

10.1.1 Die „eigenen Dateien“ sichern
Sie wollen schnell einmal ihre Dokumente, Bilder, Videos und musik sichern? Öffnen Sie die eingabe-
aufforderung (als normalnutzer, nicht als Administrator, sonst wird viel unnützes Zeug kopiert). Falls Sie
ihre Daten nicht auf andere Laufwerke verlagert haben, wie im Kapitel 7.2 empfohlen, befinden sie sich in
den unterordnern von C:\users\<Benutzername>. tippen Sie den folgenden Befehl ein und drücken
Sie „enter“:

robocopy %userprofile% e:\backup /s /r:0

Windows ersetzt die Variable %userprofile%

durch C:\users\<Benutzername> und legt
auf Laufwerk E: einen Ordner „Backup“ als
Zielverzeichnis an. Dann beginnt das Kopieren.
Sie können dabei zusehen und ungefähr ver-
folgen, welche Ordner kopiert werden. 

hinweis 1: eine über den Bildschirm rasende
Ausgabe kann man mit der taste Pause oder mit
Strg-num anhalten, Fortsetzen ist mit beliebiger
taste möglich. 

hinweis 2: Klicken Sie (am Besten noch vor diesem Versuch) mit der
rechten maustaste auf die titelleiste des Fensters der eingabeaufforderung
und wählen Sie „eigenschaften“ → „Layout“. Ändern Sie die „höhe“ der
Fensterpuffergröße auf 2000 oder mehr, damit Sie in einem langen Protokoll
zurückrollen können. Vergrößern Sie die „Breite“ der Fenstergröße auf
etwa 120, damit lange Zeilen nicht umgebrochen werden. nach den „OK“
sind dieses und alle zukünftigen eingabeaufforderungsfenster breiter. 

nach einigen minuten sehen Sie ein Schlussprotokoll ähnlich wie Bild 10.1.
in Bild 10.2 sehen Sie die neu angelegten unterordner von e:\Backup. 

10.1.2 Verbesserungen

Ordner ausschließen

Vielleicht haben Sie festgestellt, das der Ordner AppData viel Platz und
Zeit beansprucht? Außer dem Benutzerprofil von thunderbird (was man
besser mit dem tool MozBackup sichern sollte) sind dort normalerweise
keine nützlichen Dateien vorhanden. Zudem erhalten Sie meldungen wie
Der Prozess kann nicht auf die Datei zugreifen, da sie von einem anderen Prozess
verwendet wird.

Löschen Sie den Ordner E:\Backup (oder auch nur E:\Backup\Appdata) und wiederholen Sie den
Robocopy-Befehl mit einem angehängten /XD AppData, also

robocopy %userprofile% e:\backup /s /r:0 /xd appdata

Der Parameter /XD schließt komplette Ordner aus dem Backup aus, deshalb wird AppData nicht mit
gesichert. nach einem Leerzeichen können Sie weitere Ordner auflisten, die nicht gesichert werden sollen. 

Bild 10.1: Schlussprotokoll von ROBOCOPY (Ausschnitt)

Bild 10.2: Die kopierten Ordner



Protokoll speichern

Sie möchten das Protokoll genau untersuchen, um sicher zu sein, dass alle wichtigen Dateien gesichert
worden sind? Wenn Sie /LOG:Dateiname.txt oder /UNILOG:Dateiname.txt an den Robocopy-
Befehl anhängen, wird das Protokoll in die angegebene Datei umgelenkt. ich rate zu „unilog“, weil nur
bei nutzung des unicode-Zeichensatzes eventuelle umlaute in Dateinamen korrekt dargestellt werden. 

Sie sollten zusätzlich /np anhängen, damit die Protokolldatei nicht mit prozentualen Fortschrittsangaben
aufgebläht wird. und wenn Sie zusätzlich /TEE anhängen, wird das Protokoll sowohl auf Bildschirm als
auch in die Datei geschrieben. Beispiel:

robocopy %userprofile% e:\backup /s /r:0 /unilog:e:\Protokoll.txt /np /tee

hier ist ein Ausschnitt aus der Datei E:\Protokoll.txt:
-------------------------------------------------------------------------------

ROBOCOPY     ::     Robustes Dateikopieren für Windows                              
-------------------------------------------------------------------------------
Gestartet: Freitag, 19. März 2021 12:43:31
Quelle : D:\Users\Klaus\
Ziel : e:\backup\
Dateien : *.*

Ausgeschl. Verzeichnisse: D:\Users\Klaus\AppData
Optionen: *.* /TEE /S /E /DCOPY:DA /COPY:DAT /PURGE /MIR /NP /R:0 /W:30 

------------------------------------------------------------------------------
..... hier habe ich einen großen Abschnitt ausgelassen ....

Neues Verz.       1 D:\Users\Klaus\Videos\
Neue Datei 504 desktop.ini

Neues Verz.       0 D:\Users\Klaus\Videos\AnyDesk\
Neues Verz.       1 D:\Users\Klaus\Videos\Captures\

Neue Datei 190 desktop.ini
Neues Verz.       0 D:\Users\Klaus\Vorlagen\

2021/03/19 12:43:38 FEHLER 5 (0x00000005) Zielverzeichnis wird mit Zeitstempel vers
ehen e:\backup\Vorlagen\
Zugriff verweigert
------------------------------------------------------------------------------

Insgesamt   KopiertÜbersprungenKeine Übereinstimmung  FEHLER    Extras
Verzeich.:        42        28         1         0               13         0
Dateien:        67        64         0         0                3         0
Bytes:  204.55 m  201.87 m         0         0           2.67 m         0
Zeiten:   0:00:07   0:00:06                              0:00:00   0:00:01

Geschwindigkeit:            31793988 Bytes/Sek.
Geschwindigkeit:            1819.266 Megabytes/Min.

Beendet: Freitag, 19. März 2021 12:43:38

Beachten Sie bitte: Der Speicherort für die Protokolldatei sollte außerhalb von Quell- und Zielverzeichnis
liegen. Achten Sie darauf, dass Sie im gewählten Ordner Schreibrecht haben.

unterbrechen und neu starten

Wenn Sie beim Beobachten des vorbeihuschenden Protokolls etwas unerwünschtes sehen (z. B. das Ordner
gesichert werden, die nicht gesichert werden brauchen), können Sie jederzeit den Befehl mit der tasten-
kombination Strg-c abbrechen. Wenn Sie die Cursortaste „Pfeil nach oben“ drücken, wird der letzte Befehl
der eingabeaufforderung zurückgeholt (Sie können mit den Cursortasten alle früheren Befehle auswählen).
Sie können den Befehl ergänzen, beispielsweise indem Sie die Liste der ausgeschlossenen Ordner ergänzen.
Führen Sie dann den Befehl erneut aus. Da Robocopy die im vorhergehenden Durchgang kopierten Dateien
nicht erneut kopiert, wird die Abbruchstelle ganz schnell erreicht und das Kopieren wird fortgesetzt.  

10.1.3 Das gesamte Systemlaufwerk sichern
mit dem Robocopy-Befehl kann man auch das gesamte Systemlaufwerk sichern, zum Beispiel:
robocopy c:\ e:\backup_c /s /r:0 /xd c:\windows "c:\program files" 

Dieser Befehl verzichtet auf das Sichern der Ordner c:\windows und c:\program files und deren
unterordner. Auch der Ordner "c:\program files (x86)" für die älteren 32-Bit-Anwendungen braucht
nicht gesichert werden. erstens ist es unwahrscheinlich, dass sich irgendwelche ihrer Daten dort befinden.
Zweitens wäre das komplette Kopieren dieser Ordner eine Verschwendung von viel Backup-Speicherplatz.
Beachten Sie die Anführungszeichen, die immer dann notwendig sind, wenn ein Datei- oder Ordnername
Leerzeichen enthält. 

eine andere möglichkeit, Speicherplatz auf dem Backup-massenspeicher zu sparen, bietet der Parameter
/xf *.exe *.dll. Damit werden die meisten ausführbaren Daten aus der Sicherung ausgeblendet.
Programme zu sichern ist ohnehin sinnlos: Wenn man Windows neu installieren muss, dann müssen auch
alle Anwendungen neu installiert werden, um die notwendigen einträge für die Registry zu erstellen. 

Der obige /xd Parameter schließt in den Windows- und Programmordnern bereits viele *.exe und
*.dll Dateien aus. Ob die gleichzeitige nutzung der /xd und /xf Parameter viel ersparnis bringt,
sollte man gegebenfalls testen. 

10.2 DAS WiChtigSte üBeR VARiABLen

Zwei Seiten vorher wurde bereits die Variable %userprofile% verwendet. Was hat es damit auf sich? 

DOS und Windows haben einen kleinen Spezialspeicher, den „umgebungsspeicher“, engl. „environment“,
um einige wichtige Parameter in textform bereitzuhalten. Öffnen Sie die „eingabeaufforderung“, geben Sie
den Befehl SET ein und schauen Sie das ergebnis an! Sie finden unter anderem den netzwerknamen
COMPUTERNAME, ihren Anmeldenamen USERNAME und die Ordnernamen SystemRoot, WINDIR und
ProgramFiles. Sehr nützlich ist die Variable USERPROFILE, die die persönlichen Datenordner anzeigt.
Diese Variablen können von Programmen und auch von Stapeldateien benutzt werden. Auf der nächsten
Seite werden die Variablen DATE und TIME vorgestellt, deren Wert von Windows aktuell gehalten wird. 

im umgebungsspeicher ist etwas Platz übrig für selbstdefinierte Variablen. mit einem Befehl SET TEST=JA
wird eine Variable TEST erzeugt und sie bekommt den Wert „JA“ zugewiesen. mit SET können Sie das
überprüfen. mit SET TEST=NEIN kann man den Wert ändern. Die Variable kann man in Programmen
nutzen. Bei Verwendung in einer Stapeldatei muss man die Variable mit Prozentzeichen umschließen,
dann wird sie während der Ausführung des Programms durch ihren Wert ersetzt. 

Wird die Variable nicht mehr benötigt, sollte sie mit SET TEST= aus dem umgebungsspeicher entfernt
werden, um den belegten Speicherplatz freizugeben. Achtung, sofort hinter dem gleichheitszeichen enteR
drücken, es darf kein Leerzeichen folgen!

Ändern Sie nicht die Werte der Systemvariablen (USERNAME, WINDIR u. a.), die Folgen wären unabsehbar! 

nach einer neuinstallation enthält die Variable COMPUTERNAME wahrscheinlich eine Zufallszahl und ist
nicht besonders hilfreich. Den Computernamen können Sie ändern. unter Windows 10 gehen Sie über
„einstellungen“ → „System“ → „info“ zu „Diesen PC umbenennen (fortgeschritten), unter Windows 7 und
bei älteren Versionen von Windows 10 kommen Sie mit der tastenkombination Windows-taste plus „Pause“-
taste zum Fenster „System“. Bei den „einstellungen für den Computernamen, ...“ wird der aktuelle name
angezeigt. Klicken sie auf „einstellungen ändern“. 

nach „Ändern“ geben Sie einen selbsterklärenden namen ein (max. 15 alphanumerische Zeichen, keine
umlaute, keine Sonderzeichen außer dem minuszeichen). Klicken Sie auf „OK“. geben Sie eine „Computer-
beschreibung“ ein, wenn Sie wollen. Schließen Sie alle Fenster mit „OK“ und führen Sie einen neustart aus. 

Beachten Sie: Wenn andere netzwerkgeräte eine ständige Verbindung mit ihrem PC hatten, geht diese
Verbindung durch eine namensänderung verloren und muss erneut hergestellt werden. 

10.2.1 Die Variablen DAte und time
machen Sie mit bei einem kleinen, ungefährlichen Versuch. Öffnen Sie die eingabeaufforderung, indem
Sie „cmd“ in das Suchfeld eintippen, gefolgt von einem Druck auf die „enter“-taste (wie das geht, steht
ausführlicher im Anhang 16.1.1). geben Sie den folgenden Befehl ein, gefolgt von enter:

time

Sie erhalten eine meldung, die etwa so aussieht:

Aktuelle Zeit: 14:04:55,84

Geben Sie die neue Zeit ein: 

Sie könnten jetzt die uhrzeit korrigieren, falls sie nicht stimmt. Wenn die angezeigte Zeit korrekt ist, drücken
Sie nur die enter-taste.

Auch das Datum können Sie abfragen. geben Sie den folgenden Befehl ein, gefolgt von enter:

date

Sie erhalten eine meldung, die etwa so aussieht:

Aktuelles Datum: 30.12.2020

Geben Sie das neue Datum ein: (TT-MM-JJ) 

Sie könnten jetzt das Datum eingeben, falls es nicht stimmt. Beachten Sie die vorgeschriebene Schreibweise.
Wenn Sie mit dem angezeigten Datum einverstanden sind, drücken Sie nur die enter-taste.

Je nach Windows-Version und -einstellungen kann die meldung zusätzlich den Wochentag enthalten:

Aktuelles Datum: Sa 30.12.2020 

Geben Sie das neue Datum ein: (TT-MM-JJ) 

Kennen Sie den Befehl „echo“? er wird hauptsächlich in Stapeldateien zur textausgabe auf dem Bildschirm
verwendet. Probieren Sie einmal:

echo %date% 

Je nach Windows-Version erhalten Sie 30.12.2020

oder Mi,30.12.2020

merken Sie sich, welche der beiden Datumsversionen für ihr Betriebssystem zutrifft: mit oder ohne
Wochentag. 

in den Beispielen auf den folgenden Seiten steht Q: für das Quellen-Laufwerk mit den zu sichernden
Daten und Z: für das Ziel-Laufwerk, wohin die Daten gesichert werden sollen. ersetzen Sie Z: durch
den Laufwerksbuchstaben, welchen Windows dem uSB-Stick oder der externen Festplatte zugewiesen hat.
Wenn Sie diesen Buchstaben nicht wissen, ermitteln Sie ihn mit dem Windows explorer. Wenn Sie Admi-
nistratorrechte haben, können Sie auch den Festplattenmanager (Befehl diskmgmt.msc an der ein-
gabeaufforderung) benutzen, um die Laufwerksbuchstaben zu ermitteln. 

Q: müssen Sie vermutlich durch C: oder D: ersetzen oder wo auch immer Sie ihre Daten gespeichert
haben. 

Die Variable %DATE% enthält das aktuelle Datum. mit dem Befehl 

md Z:\%DATE% 

kann man an jedem tag ein Verzeichnis erstellen, das so heißt wie das Datum. Probieren Sie es einfach
einmal aus: geben Sie den Befehl ein und überzeugen Sie sich, dass im Ziel-Laufwerk ein neues Verzeichnis
entstanden ist. Wenn Sie anschließend den Befehl

xcopy Q:\*.doc* Z:\%DATE%\*.doc* /s oder

robocopy Q:\ Z:\%DATE%\ *.doc* /s /r:0 

eingeben und enter drücken, werden alle ihre Word-Dokumente in dieses neue tagesverzeichnis kopiert.

ein nachteil dieses Verfahrens: Bei einer täglichen Datensicherung würden pro Jahr 365 tagesverzeichnisse
entstehen, von denen jedes viele identische Dateien enthalten würde. 365 ähnliche Kopien würden wohl
kaum auf die Festplatte passen. es wäre sinnvoller, nur einen teil des Datums zu verwenden. Wenn man
vom Datum beispielsweise nur die nummer des monats verwendet, erhält man jeden monat einen neuen
Ordner, insgesamt 12 Ordner pro Jahr. 

Wenn das Datumsformat des PC keinen vorangestellten Wochentag aufweist, wird die Variable
%DATE:~3,2% die ersten drei Stellen aus dem Datum ignorieren und die darauffolgenden zwei Zeichen
herausschneiden: Das ist die nummer des monats. Falls das Datumsformat auf ihrem PC einen vorange-
stellten Wochentag aufweist, müssen Sie den Parameter %DATE:~6,2% verwenden. Jedoch ist eine Daten-
sicherung, die je nach Betriebssystem unterschiedlich funktioniert, äußerst unpraktisch. 

Doch es gibt einen Ausweg: Wir zählen die benötigten Datumsbestandteile von hinten ab. in %DATE:~-

7,2% bewirkt das minuszeichen, dass von hinten abgezählt wird! Sieben Zeichen vom ende beginnend
werden zwei Zeichen herausgeschnitten, die nummer des monats. %DATE:~-4,4% oder kürzer
%DATE:~-4% ergibt die vierstellige Jahreszahl, und %DATE:~-10,2% liefert den tag des monats. Probieren
Sie mit echo %DATE:~-7,2% verschiedene Parameter aus! 

Wenden wir das gelernte an. Wenn Sie die folgenden beiden Befehle

md Z:\%DATE:~-7,2% 
xcopy Q:\*.doc* Z:\%DATE:~-7,2%\*.doc*  /s 

regelmäßig ausführen, erhalten Sie im Laufe des Jahres 12 Ordner für ihre Word-Dateien. egal, ob Sie Daten
mit diesem Befehl täglich, wöchentlich oder gelegentlich sichern, in jedem Fall erfolgen alle Sicherungen
in den Ordner des aktuellen monats. Jeden monat wird ein neuer Ordner begonnen. Allerdings wird nach
einem Jahr das monatsverzeichnis vom Vorjahr mit den Daten des aktuellen Jahres überschrieben. Wenn Sie
das nicht wünschen, müssen Sie den namen des monatsordners aus monats- und Jahresnummer zusammen-
setzen. %DATE:~6,4% liefert die Jahreszahl. Die Befehle 

md Z:\%DATE:~-4%-%DATE:~-7,2% 

xcopy Q:\*.* Z:\%DATE:~-4%-%DATE:~-7,2%\*.*  /s 

würden am 30.12.2020 einen Ordner mit dem namen 2020-12 erstellen. und weil als Dateiname
*.* angegeben ist, werden alle Dateien gesichert, nicht wie im vorhergehenden Beispiel nur die
Word-Dokumente. 

Datensicherung mit nutzung der DAte-Variable

Jetzt nutzen wir die DAte-Variable für eine Datensicherung. mit dem Befehl

robocopy c\ e:\backup_c\%DATE:~-7,2% /s /r:0 /xf *.exe *.dll

sichern Sie das ganze Laufwerk C: mit Ausnahme der Programmdateien *.exe und *.dll, und mit

robocopy d\ e:\backup_d\%DATE:~-7,2% /s /r:0 /xf *.exe *.dll

tun Sie dasselbe für Laufwerk D:. Wobei die Sicherung jeden monat in einen anderen Ordner erfolgt, ins-
gesamt zwölf Ordner. nun müssen Sie nur nach daran denken, diese Befehle öfter einmal auszuführen.
Das wäre am bequemsten, wenn Sie eine Stapeldatei mit diesen beiden Befehlen erstellen (und als dritten
Befehl „pause“ anhängen, damit Sie das ergebnis der Ausführung sehen können). 

10.2.2 Zugriffsrechte

Was sind Zugriffsrechte? 

Die meisten Dateien auf der Festplatte sind für jedermann zugänglich, mit Ausnahme von einigen system-
kritischen Dateien und Ordnern, die speziell geschützt sind. eine weitere Ausnahme: Auf dem Windows-
Systemlaufwerk gibt es einen Ordner C:\User\ bzw. C:\Benutzer\ und darin gibt es für jeden
Benutzer einen unterordner, auf den nur der jeweilige Benutzer Zugriff hat. Beispielsweise hat der Benutzer
mit dem Anmeldenamen Klaus einen eigenen Ordner C:\User\Klaus\, der wiederum die unterordner
C:\User\Klaus\Documents, C:\User\Klaus\Downloads und andere hat. Diese eigenen Ordner
sind für alle anderen Benutzer gesperrt. Beim Versuch, auf einen „fremden“ Ordner zuzugreifen, sehen sie
die folgende Fehlermeldung:

„Sie verfügen momentan nicht über die Berechtigung des Zugriffs auf diesen Ordner.
Klicken Sie auf „ Fortsetzen “, um dauerhaft Zugriff auf diesen Ordner zu erhalten.“

Wenn Sie auf „Fortsetzen“ klicken, hängt das ergebnis davon ab, ob Sie über Administratorrechte verfügen.
Als Benutzer ohne Administratorrechte erhalten Sie das Angebot, durch eingabe des Administratorpassworts
zeitweilig zum Administrator hochgestuft zu werden. mit den Administratorrechten wird dann der Zugriff
auf den fremden Ordner eingerichtet, was durchaus eine minute dauern kann. Wichtig: Auch ein Benutzer
mit Administratorrechten bekommt erst dann Zugriff, nachdem er diesen angefordert hat! 

Das Problem der Zugriffsrechte bei der Datensicherung

Daraus folgt, dass zunächst einmal kein Benutzer die Datensicherung für einen anderen Benutzer durch-
führen kann. Schlimmer noch: ein Benutzer kann die Sicherung anderer Benutzer zerstören. nehmen wir
an, es gibt zwei Benutzer A und B. A kann die Daten von B nicht sehen und umgekehrt. Wenn nun Benutzer
A die Datensicherung durchführt, sichert er die allgemein zugänglichen Dateien und auch seine eigenen
Daten, doch die von B kann er nicht sehen und folglich auch nicht sichern. 

Wenn anschließend nutzer B die Datensicherung durchführt, werden seine eigenen Daten gesichert. Doch
weil B die Daten von A nicht sehen kann – anders ausgedrückt: einen leeren Ordner sieht, wo sich die
Daten von A befinden – bringt ROBOCOPY die Ordner in übereinstimmung und löscht die vom nutzer A
gesicherten Daten aus dem Backup heraus (die Originaldaten von A sind aber nicht in gefahr). 

gegen dieses Dilemma gibt es vier möglichkeiten:

1. Die einfachste Lösung: Jeder Benutzer gibt demjenigen Benutzer, der die Sicherung durchführen soll,
das Vollzugriffsrecht oder zumindest das Leserecht für seine Dateien. 

2. ein Benutzer mit Administratorrechten sichert die Daten und verschafft sich vorher die Zugriffsrechte.

3. Jeder Benutzer sichert nur seine eigenen Daten auf einen eigenen, persönlichen Datenträger oder in
einen eigenen Bereich auf dem gemeinsamen Sicherungs-Datenträger. Darüber hinaus muss jemand
die gemeinsam genutzten Dateien sichern. 

4. Sie verzichten im Robocopy-Befehl auf den Parameter „/PuRge“. Doch das hat zwei nachteile:

• Das führt zu einer gewissen Verschwendung von Speicherplatz. Wenn ein nutzer einige Ordner
an einen neuen Platz verschiebt, werden die Dateien am neuen Platz gesichert, die Kopie der
Daten am alten Speicherplatz wird jedoch nicht gelöscht. 

• Jeder nutzer muss sich selbst um die Sicherung seiner Daten kümmern oder einen Administrator
darum bitten.

Die ersten vier möglichkeiten sind zu empfehlen, die fünfte nicht. 

Wie werden Zugriffsrechte vergeben?

Lokale Zugriffsrechte können Sie sowohl
für Ordner als auch für Dateien vergeben.
Klicken Sie im explorer mit der rechten
maustaste auf den Ordner (bzw. die Datei).
Öffnen Sie die „eigenschaften“ und wählen
Sie die Registerkarte „Sicherheit“. im Bild
werden die Berechtigungen für Benutzer
„Klaus“ angezeigt. 

um Rechte an einem weiteren Benutzer zu
vergeben, klicken Sie auf „Bearbeiten“ und
„hinzufügen“. im Bild 10.4 schreiben Sie
den namen des Benutzers ein. ein Klick
auf „namen überprüfen“ und „OK“ trägt
den neuen Benutzer ein, siehe Bild 10.5,
zunächst nur mit minimalen Rechten. 

Welche Rechte können vergeben werden?

Vollzugriff: erlaubt Benutzer Dateien und unterverzeichnisse
zu lesen, schreiben, ändern und löschen. Außerdem können
Benutzer die Berechtigungseinstellungen für alle Dateien und
unterverzeichnisse ändern.

Ändern: erlaubt Benutzern, Dateien und unterverzeichnisse
zu lesen und zu schreiben; auch das Löschen des Ordners ist
möglich.

Lesen, Ausführen: erlaubt Benutzern, ausführbare Dateien inklusive
Skripte anzuzeigen und auszuführen.

Ordnerinhalt anzeigen: erlaubt das Anzeigen und Auflisten von
Dateien und unterverzeichnissen sowie das Ausführen von Dateien;
wird nur auf Ordner vererbt.

Lesen: erlaubt Benutzern, die inhalte von Ordnern und unterver-
zeichnissen anzuzeigen.

Schreiben: erlaubt Benutzern, Dateien und unterverzeichnisse hin-
zuzufügen, erlaubt ihnen, in eine Datei zu schreiben.

Bild 10.3: Standardrechte des Benutzers Klaus

Bild 10.4: Benutzer Alina hinzufügen

Bild 10.5: minimale Rechte für Alina



10.3 DAtenSiCheRung (FASt) Ohne PROtOKOLLieRung

10.3.1 einfachste Version für einen PC
Wenn Sie Daten häufig oder regelmäßig sichern wollen, ist es sinnvoll, für diesen Vorgang eine Stapeldatei
zu erstellen. Die folgende Stapeldatei sichert ein oder zwei Laufwerke (z. B. C: und D:) auf eine externe
Festplatte (z. B. F:). Jeden monat wird ein neuer Ordner für das Backup angelegt. mit dieser Stapeldatei
ist es nicht möglich, zwei oder mehr PC auf eine gemeinsam genutzte Backup-Festplatte zu sichern. 

nach dem ende der Sicherung wird der Start- und ende-Zeitpunkt angezeigt. Daraus kann man in der Regel
erkennen, ob die Sicherung zu einem normalen ende gekommen ist. 

Sie wollen diese Datei ausprobieren? Sie finden auch diese die Stapeldatei auf www.eifert.net im
Begleitmaterial zum Buch „Datensicherung für Anfänger“. 

Alle Zeilen, die zur Stapeldatei gehören, sind blau oder rot gefärbt. Die rot gefärbten Befehle PAuSe sind
sinnvoll während der testphase, um die korrekte Ausführung der Befehle Schritt für Schritt kontrollieren zu
können. nach der testphase sollten die roten Pause-Befehle entfernt werden, damit das Programm unbe-
aufsichtigt bis zum ende durchläuft. 

Zeilen, die mit Rem (Remarque) beginnen, sind Kommentare, die in der Stapeldatei bleiben sollten.

Kursive textzeilen in Schwarz sind erläuterungen, die nur hier im Buch zu finden sind. . 

Sehr wichtig: Diese Stapeldatei darf keinesfalls auf der internen Festplatte des PC gespeichert werden,
sondern sie muss sich auf der externen Festplatte (hier: Laufwerk F:) befinden! 

nehmen wir an, es wäre der 30.12.2020, 18:00 uhr und der PC heißt ASuS-PC. Außerdem nehmen wir
an, dass es zwei am 01.05.2018 durchgeführte vollständige Datensicherungen gibt, die auf verschiedene
massenspeicher erfolgten, so dass Daten vor diesem tag nicht mehr gesichert werden brauchen. 

Die Variable mAXAge bekommt als Wert das Datum 01.05.2018 zugewiesen. Alle Dateien, deren 
letzte Änderung vor diesem Tag liegt, werden ignoriert. Ändern Sie diesen Wert nach Ihrem Bedarf,
z. B. auf das Datum der letzten gut geprüften, doppelt vorhandenen Vollsicherung. 

SET MAXAGE=20180501

In der Kommentarzeile wird erläutert, welchen Wert die folgenden Variablen annehmen dürfen. In 
diesem Beispiel sollen die Laufwerke C: und D: nach F: gesichert werden. 

rem QLWx = Quelle (ohne Doppelpunkt), ZLW=Ziel (mit Doppelpunkt)

SET QLW1=C

SET QLW2=D

SET ZLW=F:

rem Aus dem Datum werden die Variablen JAHR, MONAT und TAG ausgeschnitten

set JAHR=%date:~-4%

set MONAT=%date:~-7,2%

set TAG=%date:~-10,2%

Wäre heute der 30.12.2020, so wären die Variablen TAG=30, MONAT=12 und JAHR=2020.
echo Kontrollausgabe: Es ist der %TAG%. Tag des %MONAT%. Monats im Jahr %JAHR%

Jetzt beginnt das Erstellen der Ordner und der Kopiervorgang.
echo Bitte einige Minuten Geduld, waehrend die Festplatte durchsucht wird!

Der Befehl titLe schreibt den dahinter stehenden Text in die Kopfzeile des Fensters der
Eingabeaufforderung, um den Benutzer über den Programmfortschritt zu informieren.

TITLE Das Monatsverzeichnis wird synchronisiert

Auf dem Ziellaufwerk (der externen Festplatte) wird eine Ordnerstruktur angelegt, bestehend aus
Jahr, Monat und dem Laufwerksbuchstaben, z. B. F:\2020-12\C.

Falls es die Ordner schon gibt, sehen Sie eine Fehlermeldung – ignorieren Sie diese. 
md %ZLW%\%JAHR%-%MONAT% 

md %ZLW%\%JAHR%-%MONAT%\%QLW1%

Der folgende Echo-Befehl erzeugt einen Text „Start am 30.12.2020 um 18:00 uhr“. 
Der Text würde normalerweise auf den Bildschirm ausgegeben, doch der Umleitungspfeil „>“ 
leitet die Textausgabe in eine Datei um. Der vom Echo-Befehl erzeugte Text wird in eine
Datei DAueR.tXt im Ordner F:\ umgeleitet.

echo Start am %DATE% um %TIME:~0,5% Uhr. >%ZLW%\DAUER.TXT

PAUSE

Jetzt folgt der Kopierbefehl. Seine vielen Parameter werden auf der nächsten Seite erklärt. Das 
Einsetzen der Werte in die Variablen ergibt
ROBOCOPY C:\ F:\2020-12\C\ /S /PURGE /FP /maxage:20180501 /XA:RSH ... usw. 

ROBOCOPY %QLW1%:\ %ZLW%\%JAHR%-%MONAT%\%QLW1%\ /S /PURGE /FP

/maxage:%MAXAGE% /XA:RSH /XD %WINDIR% "%SystemDrive%\Program Files (x86)"
"%ProgramFiles%" RECOVER RECYCLER $Recycle.Bin "System Volume Information"

/XF *.tib *.img *.msi *.tmp *.exe *.dll *.com *.ini ntuser.* /r:0

Der folgende Echo-Befehl erzeugt einen Text „ende Sicherung C: um 18:20 uhr. 
Diese Nachricht würde normalerweise auf dem Bildschirm erscheinen. Ein einfacher Umleitungs-
pfeil „>“ würde die Nachricht in eine neue erstellte Datei schreiben. Der doppelte Umleitungs-
pfeil „>>“ leitet die Textausgabe in eine vorhandene Datei um, ohne diese zu überschreiben. Die
Nachricht wird an die Datei DAueR.tXt im Ordner F:\2020-12\ angehängt.

echo Ende Sicherung %QLW1%: um %TIME:~0,5% Uhr. >>%ZLW%\DAUER.TXT

PAUSE

rem Pruefen, ob ein zweites Laufwerk gesichert werden soll

Es wird geprüft, ob die Zeichenfolgen "%QLW2%x" und "x" identisch sind. Wenn ja, wurde für 
das zweite Laufwerk kein Wert eingetragen. In diesem Fall werden die nachfolgenden Befehle 
übersprungen und ab der Marke :SICHTBAR wird die Programmabarbeitung fortgesetzt.

if "%QLW2%x" == "x" goto SICHTBAR

es ist ein weiteres Laufwerk zu sichern. 
TITLE Die zweite Partition %QLW2%: wird gesichert

Ein weiterer Ordner wird angelegt, z. B. F:\2020-12\D

md %ZLW%\%JAHR%-%MONAT%\%QLW2%

Der zweite Kopierbefehl ist identisch mit dem ersten, es wird nur QLW1 durch QLW2 ersetzt.
ROBOCOPY %QLW2%:\ %ZLW%\%JAHR%-%MONAT%\%QLW2%\ /S /PURGE /FP

/maxage:%MAXAGE% /XA:RSH /XD %WINDIR% "%SystemDrive%\Program Files (x86)"
"%ProgramFiles%" RECOVER RECYCLER $Recycle.Bin "System Volume Information"

/XF *.tib *.img *.msi *.tmp *.exe *.dll *.com *.ini /r:0

PAUSE

echo Ende Sicherung %QLW2%: um %TIME:~0,5% Uhr. >>%ZLW%\DAUER.TXT

Der vorangestellte Doppelpunkt macht SICHTBAR zu einer „Sprungmarke“, zum potenziellen Ziel
eines GOTO-Befehls. Dadurch kann der zweite Kopierbefehl übersprungen werden.

:SICHTBAR

@echo off deaktiviert die Anzeige der Befehle, damit das Protokoll von ROBOCOPY sichtbar bleibt.
@echo off

Unter Umständen sind die kopierten Daten unsichtbar. Dieser Attrib-Befehl macht sie sichtbar.
attrib -h -s %ZLW%\*.* /s /d

rem Hier koennten Befehle folgen, die ueber das Ende der Sicherung informieren

Die Variablen mOnAt, tAg und JAhR werden nicht mehr benötigt und werden deshalb aus dem
Umgebungsspeicher gelöscht. Beachten Sie: Hinter dem Gleichheitszeichen darf nichts stehen, auch
kein Leerzeichen, sonst werden die Variablen nicht gelöscht, sondern mit neuem Inhalt gefüllt.

set JAHR=
set MONAT=
set TAG=
:SCHLUSS
type %ZLW%\DAUER.TXT
pause

Die Stapeldatei ist beendet. 
Die „Eingabeaufforderung“ ist geöffnet und man kann die letzten Meldungen der Stapeldatei sehen.
Als letztes wird die Datei DAUER.TXT im Ordner %ZLW%\ angezeigt:

Start am 30.12.2020 um 18:00 Uhr.
Ende Sicherung C: um 18:20 Uhr.
Ende Sicherung D: um 18:26 Uhr.

in der Datei sind vier PAuSe-Befehle enthalten. Wenn Windows auf einen PAuSe-Befehl trifft, stoppt die
Abarbeitung, bis Sie eine taste drücken. nutzen Sie diese Zwischenstopps in der testphase, um mit dem
explorer nachzusehen, welche Ordner und Dateien erstellt worden sind und wie die Protokolldateien aus-
sehen. Fügen Sie gegebenfalls weitere PAuSe-Befehle an Stellen ein, wo Sie den Programmablauf genauer
untersuchen wollen. Wenn alles zu ihrer Zufriedenheit läuft, entfernen Sie alle PAuSe-Befehle (oder
schreiben Sie ein „REM “ davor). Der letzte Pause-Befehl darf nicht entfernt werden, weil Sie sonst keine
Rückmeldung haben, ob die Sicherung zu einem normalen ende gekommen ist. 

übrigens: Warum werden im zweiten Robocopy-Befehl, beim Sichern von Laufwerk D: die Windows-und
Programmordner ausgeschlossen? Weil mancher meiner Kunden auf einem Laufwerk Windows 7 und auf
dem anderen Laufwerk Windows 10 installiert hat. 

10.3.2 einfachste Version für mehrere PC
Die soeben vorgestellte Stapeldatei ist nicht geeignet für PCs, die eine gemeinsame Backup-Festplatte nutzen
sollen. Denn jeder PC hat ein Laufwerk C:, und die Sicherung jedes PC würde die Sicherung anderer PC
überschreiben. Deshalb zeige ich ihnen die Stapeldatei von 10.3.1 noch einmal − ergänzt nur um den
Computernamen. Dadurch bekommt jeder PC seinen eigenen Ordner mit eigenen unterordnern. 

Die verbesserte Stapeldatei ist universell in dem Sinne, dass diese Stapeldatei mehrere PCs nacheinander
auf eine gemeinsam genutzte große Backup-Festplatte kopieren kann. Je nachdem, welche unterschiede
zwische ihren PC existieren, müssen Sie wahrscheinlich für jeden PC eine individualisierte Kopie erstellen. 

Was müssen Sie vor der Benutzung anpassen? 

• Die Laufwerksbuchstaben: Wenn alle ihre PC nur Laufwerk C: und D: haben, bleibt es bei Set
QLW1=C und Set QLW2=D. Wenn ein PC kein Laufwerk D: hat, funktioniert die Stapeldatei trotzdem,
es wird dann nur Laufwerk C: gesichert. 

• Der Laufwerksbuchstabe der Backup-Festplatte: er ist möglicherweise an jedem PC anders. Doch Sie
können mit dem Festplattenmanager auf jedem PC den gleichen Laufwerksbuchstaben für die Backup-
Festplatte festlegen.

• Legen Sie mit mAXAge fest, wie weit die Sicherung in die Vergangenheit zurückgreifen soll. 

übrigens: Die Backup-Festplatte muss nicht leer sein, vorhandener inhalt bleibt erhalten. Die Stapeldateien
in diesem Kapitel 10 löschen nichts, sie legen nur neue Ordner an, die wie das Jahr oder wie der Computer
heißen. und es wird eine Datei DAUER.TXT angelegt. 

nehmen wir an, es wäre der 30.12.2020, 18:00 uhr und der PC heißt ASuS-PC. Außerdem nehmen wir
an, dass es zwei am 01.05.2018 durchgeführte vollständige Datensicherungen gibt, die auf verschiedene
massenspeicher erfolgten, so dass Daten vor diesem tag nicht mehr gesichert werden brauchen. 

Die Variable mAXAge bekommt als Wert das Datum 01.05.2018 zugewiesen. Alle Dateien, deren 
letzte Änderung vor diesem Tag liegt, werden ignoriert. Ändern Sie diesen Wert nach Ihrem Bedarf,
z. B. auf das Datum der letzten gut geprüften, doppelt vorhandenen Vollsicherung. 

SET MAXAGE=20180501

In der Kommentarzeile wird erläutert, welchen Wert die folgenden Variablen annehmen dürfen. In 
diesem Beispiel sollen die Laufwerke C: und D: gesichert werden. Das Ziellaufwerk (die externe
Festplatte) ist F:

rem QLWx = Quelle (ohne Doppelpunkt), ZLW=Ziel (mit Doppelpunkt)

SET QLW1=C

SET QLW2=D

SET ZLW=F:

rem Aus dem Datum werden die Variablen JAHR, MONAT und TAG ausgeschnitten

set JAHR=%date:~-4%

set MONAT=%date:~-7,2%

set TAG=%date:~-10,2%

Wäre heute der 30.12.2020, so wären die Variablen TAG=30, MONAT=12 und JAHR=2020.
echo Kontrollausgabe: Es ist der %TAG%. Tag des %MONAT%. Monats im Jahr %JAHR%

Jetzt beginnt das Erstellen der Ordner und der Kopiervorgang.

echo Bitte einige Minuten Geduld, waehrend die Festplatte durchsucht wird!

Der Befehl titLe schreibt den dahinter stehenden Text in die Kopfzeile des Fensters der
Eingabeaufforderung, um den Benutzer über den Programmfortschritt zu informieren.

TITLE Das Monatsverzeichnis wird synchronisiert

Auf dem Ziellaufwerk (der externen Festplatte) wird eine Ordnerstruktur angelegt: Netzwerkname
des PC, gefolgt von Jahr, Monat und dem Laufwerksbuchstaben, z. B. F:\ASUS-PC\2020-12\C.
Falls es die Ordner schon gibt, sehen Sie eine Fehlermeldung – ignorieren Sie diese. 

md %ZLW%\%COMPUTERNAME%

md %ZLW%\%COMPUTERNAME%\%JAHR%-%MONAT% 

md %ZLW%\%COMPUTERNAME%\%JAHR%-%MONAT%\%QLW1%

Der folgende Echo-Befehl erzeugt einen Text „Start am 30.12.2020 um 18:00 uhr“. 
Der Text würde normalerweise auf den Bildschirm ausgegeben, doch der Umleitungspfeil „>“ 
leitet die Textausgabe in eine Datei um. Der vom Echo-Befehl erzeugte Text wird in eine
Datei DAueR.tXt im Ordner F:\ASuS-PC\ umgeleitet.

echo Start am %DATE% um %TIME:~0,5% Uhr. >%ZLW%\%COMPUTERNAME%\DAUER.TXT

PAUSE

Jetzt folgt der Kopierbefehl. Seine vielen Parameter werden auf der nächsten Seite erklärt. Das 
Einsetzen der Werte in die Variablen ergibt
ROBOCOPY C:\ F:\ASUS-PC\2020-12\C\ /S /PURGE /FP /maxage:20180501 ... usw. 

ROBOCOPY %QLW1%:\ %ZLW%\%COMPUTERNAME%\%JAHR%-%MONAT%\%QLW1%\ /S /PURGE /FP

/maxage:%MAXAGE% /XA:RSH /XD %WINDIR% "%SystemDrive%\Program Files (x86)"
"%ProgramFiles%" RECOVER RECYCLER $Recycle.Bin "System Volume Information"

/XF *.tib *.img *.msi *.tmp *.exe *.dll *.com *.ini ntuser.* /r:0

Der folgende Echo-Befehl erzeugt einen Text „ende Sicherung C: um 18:20 uhr. 
Diese Nachricht würde normalerweise auf dem Bildschirm erscheinen. Ein einfacher Umleitungs-
pfeil „>“ würde die Nachricht in eine neue erstellte Datei schreiben. Der doppelte Umleitungs-
pfeil „>>“ leitet die Textausgabe in eine vorhandene Datei um, ohne diese zu überschreiben. Die
Nachricht wird an die Datei DAueR.tXt im Ordner F:\ASuS-PC\2020-12\ angehängt.

echo Ende Sicherung %QLW1%: um %TIME:~0,5% Uhr. >>%ZLW%\%COMPUTERNAME%\DAUER.TXT

PAUSE

rem Pruefen, ob ein zweites Laufwerk gesichert werden soll

Es wird geprüft, ob die Zeichenfolgen "%QLW2%x" und "x" identisch sind. Wenn ja, wurde für 
das zweite Laufwerk kein Wert eingetragen. In diesem Fall werden die nachfolgenden Befehle 
übersprungen und ab der Marke :SICHTBAR wird die Programmabarbeitung fortgesetzt.

if "%QLW2%x" == "x" goto SICHTBAR

es ist ein weiteres Laufwerk zu sichern. 
TITLE Die zweite Partition %QLW2%: wird gesichert

Ein weiterer Ordner wird angelegt, z. B. F:\ASUS-PC\2020-12\D

md %ZLW%\%COMPUTERNAME%\%JAHR%-%MONAT%\%QLW2%

Der zweite Kopierbefehl ist identisch mit dem ersten, es wird nur QLW1 durch QLW2 ersetzt.
ROBOCOPY %QLW2%:\ %ZLW%\%COMPUTERNAME%\%JAHR%-%MONAT%\%QLW2%\ /S /PURGE /FP

/maxage:%MAXAGE% /XA:RSH /XD %WINDIR% "%SystemDrive%\Program Files (x86)"
"%ProgramFiles%" RECOVER RECYCLER $Recycle.Bin "System Volume Information"

/XF *.tib *.img *.msi *.tmp *.exe *.dll *.com *.ini /r:0

PAUSE

echo Ende Sicherung %QLW2%: um %TIME:~0,5% Uhr. >>%ZLW%\%COMPUTERNAME%\DAUER.TXT

Der vorangestellte Doppelpunkt macht SICHTBAR zu einer „Sprungmarke“, zum potenziellen Ziel
eines GOTO-Befehls. Dadurch kann der zweite Kopierbefehl übersprungen werden.

:SICHTBAR

@echo off deaktiviert die Anzeige der Befehle, damit das Protokoll von ROBOCOPY sichtbar bleibt.
@echo off

Unter Umständen sind die kopierten Daten unsichtbar. Dieser Attrib-Befehl macht sie sichtbar.
attrib -h -s %ZLW%\%COMPUTERNAME%\*.* /s /d

rem Hier koennten Befehle folgen, die ueber das Ende der Sicherung informieren

Die Variablen mOnAt, tAg und JAhR werden nicht mehr benötigt und werden deshalb aus dem
Umgebungsspeicher gelöscht. Beachten Sie: Hinter dem Gleichheitszeichen darf nichts stehen, auch
kein Leerzeichen, sonst werden die Variablen nicht gelöscht, sondern mit neuem Inhalt gefüllt.

set JAHR=
set MONAT=
set TAG=
:SCHLUSS
type %ZLW%\%COMPUTERNAME%\DAUER.TXT
pause

Die Stapeldatei ist beendet. 
Die „Eingabeaufforderung“ ist geöffnet und man kann die letzten Meldungen der Stapeldatei sehen.
Als letztes wird die Datei DAUER.TXT im Ordner %ZLW%\%COMPUTERNAME%\ angezeigt:

Start am 30.12.2020 um 18:00 Uhr.

Ende Sicherung C: um 18:20 Uhr.

Ende Sicherung D: um 18:26 Uhr.

Von dieser Stapeldatei habe ich mehrere Kopien auf meiner externen Festplatte, für jeden PC eine andere
Variante. Sie unterscheiden sich nur in einigen der ersten Zeilen, wo in Variablen die Festlegungen gespei-
chert sind, welche Daten gesichert werden sollen. ich stecke die uSB-Festplatte an den PC, der gesichert
werden soll, und starte die zu diesem PC gehörende Stapeldatei. 

Dutzende meiner Kunden benutzen Varianten dieser Datei ebenfalls. 

10.4 Die PARAmeteR DeS ROBOCOPY-BeFehLS

Weil dieses Buch nicht 80 cm breit ist, musste ich die langen ROBOCOPY-Befehle auf je vier Zeilen
verteilen. Beim erstellen einer Stapeldatei müssen die vier Zeilen als eine Zeile eingegeben werden. (Das
bedeutet: geben Sie am ende der Zeile hinter /FP, hinter (x86)" und hinter Information" ein Leer-
zeichen ein und schreiben Sie weiter, ohne die enter-taste zu drücken!) 

/S Der Kopierbefehl betrifft auch alle unterordner (Subdirectories).

/PURGE Dateien, die seit der vorhergehenden Datensicherung aus dem Original gelöscht worden
sind, werden auch aus der Kopie gelöscht. Wenn Sie beispielsweise einen Ordner umbe
nannt haben, wird das Duplikat gelöscht und Sie sparen Speicherplatz. Doch überlegen 
Sie, ob Sie das wollen. Wenn Sie versehentlich eine Datei gelöscht haben und das nicht 
schnell genug merken, wird die Datei während des nächsten Backups von der Backup-
Festplatte gelöscht!

/FP im Protokoll werden die kopierten Dateien einschließlich Ordnernamen aufgelistet.

/maxage Dateien vor dem 01.05.2018 werden ignoriert.

/XA:RSH Dateien mit den Attributen „read only“, „system“ und „hidden“ (versteckt) werden ignoriert.

/XD Die dahinter aufgezählten Ordner werden beim Kopieren ignoriert, weil dort erfahrungs-
gemäß keine Benutzerdateien sind. Das betrifft eventuelle Recovery-Dateien, den Windows-
ordner und den bzw. die Programm-Ordner, den Papierkorb (je nach Windows-Version
hat der Papierkorb verschiedene namen), installationsdateien und Systemdateien. 

/XF Alle dahinter aufgezählten Dateiarten werden ignoriert.
*.tib und *.img sind Image-Dateien.
*.msi (Microsoft Installer) sind installationsdateien.
*.exe, *.dll, *.com und *.ini sind Programmdateien. Diese zu sichern ist
unsinnig, denn im Katastrophenfall müssen ohnehin alle Programme neu 
installiert werden. 
Obwohl die Datei ntuser.* die persönlichen einstellungen (z. B. Anmeldepass-
wort) enthält, ist deren Sicherung sinnlos, denn sie läßt sich nicht zurückkopieren. 
Sie können die Liste um weitere Dateien ergänzen, die Sie nicht sichern wollen.

/r:0 Bei Dateien, die nicht gelesen werden können, erfolgt kein zweiter Versuch. Falls ihre
Festplatte in schlechtem Zustand ist, wäre /w:1 /r:1 sinnvoll: Wenn sich eine Datei
nicht lesen lässt, erfolgt nach einer Sekunde (w:1) eine einzige (r:1) Wiederholung. Leider
dauert die Sicherung dann erheblich länger, denn Windows enthält hunderte geschützte 
oder ständig in Benutzung befindliche Dateien, die sich nicht lesen lassen. 

10.5 DAtenSiCheRung mit AuSFühRLiChem PROtOKOLL

Für ein zuverlässiges Backup braucht es zweierlei:

• es muss regelmäßig und oft genug durchgeführt werden, am Besten automatisiert.

• Jemand muss regelmäßig eine kurze Kontrolle vornehmen, ob und wie das Backup gelaufen ist. 

• Jemand muss hin und wieder genauer kontrollieren, ob alles noch immer so läuft wie geplant. 

man kann sich zur Kontrolle an den PC setzen und zusehen, wie die Sicherung abläuft. Doch man kann
Probleme kaum so schnell erkennen und darüber hinaus ist das Zeitvergeudung. Sowohl für die kurze als
auch für die ausführliche Kontrolle braucht man ein Protokoll: ein kurzes, tägliches und ein ausführliches. 

Wie bisher nehmen wir an, es wäre der 30.12.2020, 18:00 uhr und der PC heißt ASuS-PC. 
Diese Stapeldatei erzeugt die folgenden Protokolle:

• Auf F:\PC-Name\Dauer.txt wird die Startzeit und das ende jeder etappe protokolliert.

• eine Datei DATUM.TXT im Ordner F:\ASUS-PC\2020-12\ mit der information, wann im jewei-
ligen monat die letzte Datensicherung durchgeführt wurde. 

• eine Datei Protokoll.txt im Ordner F:\ASUS-PC mit der information, wann die letzte Daten-
sicherung durchgeführt wurde.

• Vier Dateien Prot1.txt, Prot2.txt, Prot3.txt und Prot4.txt im Ordner F:\ASUS-PC

mit dem ausführlichen Protokoll der letzten Datensicherung, in dem jede Datei und jeder Ordner
dokumentiert sind.

• eine Datei Protokoll2020.txt mit einem fortlaufenden Protokoll auf dem Desktop des Benutzers,
der die Sicherung ausführt, mit allen Sicherungen des Jahres (eine Zeile pro Sicherungsvorgang). 

mit diesen Protokollen ist eine Kontrolle in beliebiger Ausführlichkeit möglich. 

Die Variable mAXAge bekommt als Wert das Datum 01.05.2018 zugewiesen.
SET MAXAGE=20180501

REM QLWx = Quelle (ohne Doppelpunkt), ZLW=Ziel (mit Doppelpunkt)

SET QLW1=C

SET QLW2=D

Bis zu vier Laufwerke könnten gesichert werden, hier werden nur zwei benötigt.
SET QLW3=

SET QLW4=

SET ZLW=F:

Die eigene Variable ROBO_OPt wird in allen ROBOCOPY-Befehlen dieser Datei verwendet. Sie 
dient dazu, mit geringem Aufwand mit Parametern experimentieren zu können, ohne diese in 
allen vier ROBOCOPY-Befehlen eintragen zu müssen. 
In diesem Beispiel wird mit der Protokollierung experimentiert:
/NJS Non Job Summary: Keine abschließende Statistik
/NP No Progress: Fortschrittsanzeige (in Prozent) unterdrücken

SET ROBO_OPT=/NJS /NP

Normalerweise wird jeder Befehl angezeigt, bevor er ausgeführt wird. @echo off unterdrückt 
diese Anzeige. Der Bildschirm wird übersichtlicher. Während der Testphase sollte hier @echo on

stehen, um jeden Befehl sehen zu können.
@echo off

@rem Datumsvariablen bereitstellen

set JAHR=%date:~-4%

set MONAT=%date:~-7,2%

set TAG=%date:~-10,2%

Wäre heute der 30.12.2020, so wären die Variablen TAG=30, MONAT=12 und JAHR=2020.
echo Kontrollausgabe: Es ist der %TAG%. Tag des %MONAT%. Monats im Jahr %JAHR%

:START

echo Bitte einige Minuten Geduld, waehrend die Festplatte durchsucht wird!

TITLE Das Monatsverzeichnis wird synchronisiert

Der Anhang „2>nul“ bedeutet: Ausgabekanal 2 (der nur für Fehlermeldungen reserviert ist) wird 
ins „elektronische Nirwana“ geleitet. Konkret bedeutet es, dass eine möglicherweise zu erwartende
Fehlermeldung „Verzeichnis bereits vorhanden“ auf dem Bildschirm nicht angezeigt wird.

md %ZLW%\%COMPUTERNAME% 2>nul

md %ZLW%\%COMPUTERNAME%\%JAHR%-%MONAT% 2>nul

md %ZLW%\%COMPUTERNAME%\%JAHR%-%MONAT%\%QLW1% 2>nul

echo Start am %DATE% um %TIME:~0,5% Uhr. >%ZLW%\%COMPUTERNAME%\DAUER.TXT

ROBOCOPY %QLW1%:\ %ZLW%\%COMPUTERNAME%\%JAHR%-%MONAT%\%QLW1%\ /S /PURGE

%ROBO_OPT% /FP /maxage:%MAXAGE% /XA:RSH /XD %WINDIR% "%ProgramFiles%"
"%SystemDrive%\Program Files (x86)" RECYCLER $Recycle.Bin "System Volume
Information" /XF *.tib *.img *.msi *.tmp *.exe *.dll *.com *.ini ntuser.*

/w:1 /r:1 /TEE /unilog:%ZLW%\%COMPUTERNAME%\Prot1.txt

echo Ende Sicherung %QLW1%: um %TIME:~0,5% Uhr. >>%ZLW%\%COMPUTERNAME%\DAUER.TXT

PAUSE

rem Pruefen, ob ein Laufwerk Nr. 2 angegeben ist

Falls der Variablen QLW2 kein Wert zugewiesen wurde, lautet der Vergleich auf "x" == "x"

und die Kopierbefehle für die Laufwerke %QLW2% bis %QLW4% werden übersprungen. 
if "%QLW2%x" == "x" goto SICHTBAR

Der Ordner für den zweiten Kopiervorgang wird angelegt
md %ZLW%\%COMPUTERNAME%\%JAHR%-%MONAT%\%QLW2% 2>nul

ROBOCOPY %QLW2%:\ %ZLW%\%COMPUTERNAME%\%JAHR%-%MONAT%\%QLW2%\ /S /PURGE

%ROBO_OPT% /FP /maxage:%MAXAGE% /XA:RSH /XD %WINDIR% "%ProgramFiles%"
"%SystemDrive%\Program Files (x86)" RECYCLER $Recycle.Bin "System Volume
Information"  /XF *.tib *.img *.msi *.tmp *.exe *.dll *.com *.ini ntuser.*

/w:1 /r:1 /TEE /unilog:%ZLW%\%COMPUTERNAME%\Prot2.txt

echo Ende Sicherung %QLW2%: um %TIME:~0,5% Uhr. >>%ZLW%\%COMPUTERNAME%\DAUER.TXT

PAUSE

Ab hier gibt es Unterschiede zur „Kurzversion“ von 10.3.2. Zunächst sind zwei weitere Laufwerke zu
überprüfen. 
rem Pruefen, ob ein Laufwerk Nr. 3 angegeben ist

Falls der Variablen QLW3 kein Wert zugewiesen wurde, lautet der Vergleich auf "x" == "x"

und die Kopierbefehle für die Laufwerke %QLW3% und %QLW4% werden übersprungen. 
if "%QLW3%x" == "x" goto SICHTBAR

Der Ordner für den dritten Kopiervorgang wird angelegt
md %ZLW%\%COMPUTERNAME%\%JAHR%-%MONAT%\%QLW3% 2>nul

ROBOCOPY %QLW3%:\ %ZLW%\%COMPUTERNAME%\%JAHR%-%MONAT%\%QLW3%\ /S /PURGE

%ROBO_OPT% /FP /maxage:%MAXAGE% /XA:RSH /XD %WINDIR% "%ProgramFiles%"
"%SystemDrive%\Program Files (x86)" RECYCLER $Recycle.Bin "System Volume
Information"  /XF *.tib *.img *.msi *.tmp *.exe *.dll *.com *.ini ntuser.*

/w:1 /r:1 /TEE /unilog:%ZLW%\%COMPUTERNAME%\Prot3.txt

echo Ende Sicherung %QLW3%: um %TIME:~0,5% Uhr. >>%ZLW%\%COMPUTERNAME%\DAUER.TXT

PAUSE

rem Pruefen, ob ein Laufwerk Nr. 4 angegeben ist

if "%QLW4%x" == "x" goto SICHTBAR

md %ZLW%\%COMPUTERNAME%\%JAHR%-%MONAT%\%QLW4% 2>nul

ROBOCOPY %QLW4%:\ %ZLW%\%COMPUTERNAME%\%JAHR%-%MONAT%\%QLW4%\ /S /PURGE

%ROBO_OPT% /FP /maxage:%MAXAGE% /XA:RSH /XD %WINDIR% "%ProgramFiles%"
"%SystemDrive%\Program Files (x86)" RECYCLER $Recycle.Bin "System Volume
Information"  /XF *.tib *.img *.msi *.tmp *.exe *.dll *.com *.ini ntuser.*

/w:1 /r:1 /TEE /unilog:%ZLW%\%COMPUTERNAME%\Prot4.txt

echo Ende Sicherung %QLW4%: um %TIME:~0,5% Uhr. >>%ZLW%\%COMPUTERNAME%\DAUER.TXT

PAUSE

:SICHTBAR

Der folgende attrib-Befehl macht das versteckte Sicherungsverzeichnis sichtbar.
attrib -h -s %ZLW%\%COMPUTERNAME%\*.* /s /d

:PROTOKOLL

Der Echo-Befehl erzeugt einen Text „Die letzte Sicherung erfolgte am 30.12.2020 um 18:00 Uhr 
nach F:\ASUS-PC“ , der normalerweise auf den Bildschirm ausgegeben würde. Der Umleitungs-
pfeil „>“ leitet die Textausgabe in eine Datei um. Der vom Echo-Befehl erzeugte Text wird in eine
Datei DAtum.tXt im Ordner F:\ASuS-PC\2020-12\ umgeleitet.

echo Die letzte Sicherung erfolgte am %DATE% um %TIME:~0,5% Uhr nach

%ZLW%\%COMPUTERNAME% >%ZLW%\%COMPUTERNAME%\%JAHR%-%MONAT%\DATUM.TXT

Der nächste Befehl (1 Zeile!) prüft zunächst, ob die Variable mAXAge ein Datum (8-stellig) oder
eine Anzahl von Tagen (maximal 5-stellig) enthält. Wenn es ein Datum ist, wird an die Datei 
DAtum.tXt eine zweite Zeile angefügt mit dem Inhalt 
„Die Sicherung enthält alle Daten seit dem 01.05.2018“

if not "%MAXAGE:~5,3%"=="" echo Die Sicherung enthält alle Daten seit dem

%MAXAGE:~6,2%.%MAXAGE:~4,2%.%MAXAGE:~0,4% >>%ZLW%\%COMPUTERNAME%\%JAHR%-

%MONAT%\DATUM.TXT

Der doppelte Umleitungspfeil „>>“ im obigen Befehl bewirkt, dass die vorher erstellte Datei 
DAtum.tXt (mit der Zeile „Die letzte Sicherung erfolgte ...) nicht überschrieben wird, sondern 
dass die nächste Textzeile („Die Sicherung enthält ...“) an die Datei angehängt wird. 

Im Ordner F:\ASuS-PC\2020-12 wird eine Datei „Protokoll.txt“ erstellt, mit einer Zeile:
„Letzte Datensicherung von C: D: : : erfolgte 30.12.2020 18:00 uhr nach F:\ASuS-PC\2020-12“.

echo Letzte Datensicherung von %QLW1%: %QLW2%: %QLW3%: %QLW4%: erfolgte

%date% %time:~0,5% Uhr nach %ZLW%\%COMPUTERNAME%\%JAHR%-%MONAT%

>%ZLW%\%COMPUTERNAME%\Protokoll.txt"

if not "%mAXAge:~5,3%"=="" überprüft, ob die Variable mAXAge ein Datum oder eine Anzahl 
von Tagen enthält. Wenn es ein Datum ist, wird die Datei „Protokoll.txt“ um eine zweite Zeile 
ergänzt mit dem Inhalt: 
„Die Sicherung enthält alle Daten seit dem 01.05.2018“.

if not "%MAXAGE:~5,3%"=="" echo Die Sicherung enthält alle Daten seit dem

%MAXAGE:~6,2%.%MAXAGE:~4,2%.%MAXAGE:~0,4% >>%ZLW%\%COMPUTERNAME%\Protokoll.txt

echo Ein Jahresprotokoll wird erstellt

Der nächste Befehl (1 Zeile!) erstellt ein fortlaufendes Protokoll. Auf dem Desktop des Benutzers, 
der die Sicherung ausführt, wird eine Datei „Protokoll2020.txt“ erstellt bzw. fortgeschrieben,
mit einer Zeile pro Sicherungsvorgang. Windows ersetzt „%0“ durch den Namen der Stapeldatei. 
Weil %uSeRPROFiLe% den Benutzernamen enthält, welcher Leerzeichen enthalten könnte, muss
jeder  Dateiname, der %uSeRPROFiLe% enthält, in Anführungszeichen eingeschlossen werden. 

echo %0 sicherte %QLW1%: %QLW2%: %QLW3%: %QLW4%: am %date% durch %USERNAME%

nach %ZLW%\%COMPUTERNAME%\%JAHR%-%MONAT% 

>>"%USERPROFILE%\desktop\Protokoll%JAHR%.txt"

Nun noch aufräumen:
TITLE Überflüssige Dateien aus der Sicherung herausloeschen

Größere temporäre bzw. überflüssige Dateien werden nachträglich aus der Sicherung heraus-
gelöscht, hier als Beispiel Dateien vom Typ .nrg und .img (virtuelle Dateien von Nero und 
CloneCD).

del %ZLW%\%COMPUTERNAME%\%JAHR%-%MONAT%\TempImage.nrg /s 2>nul

del %ZLW%\%COMPUTERNAME%\%JAHR%-%MONAT%\*.img /s 2>nul

TITLE Dieses Gesamtprotokoll "Protokoll%JAHR%.txt" liegt auf Ihrem Desktop 

Das Gesamtprotokoll wird angezeigt. Wenn es Ihnen zu lang ist, kürzen Sie es hin und wieder oder 
setzen Sie ein „rem“ vor das „type“, um dessen Anzeige zu unterdrücken.

type "%USERPROFILE%\desktop\Protokoll%JAHR%.txt"

if not "%MAXAGE:~5,3%"=="" echo Die Sicherung enthaelt Daten seit dem

%MAXAGE:~6,2%.%MAXAGE:~4,2%.%MAXAGE:~0,4%

rem Hier könnten Befehle folgen, die über das Ende des Sicherung informieren

set JAHR=

set MONAT=

set TAG=

:SCHLUSS

ECHO Achtung! Je nach Zugriffsrechten sind die Daten anderer Benutzer viel-

leicht nicht gesichert! 

type %ZLW%\%COMPUTERNAME%\DAUER.TXT
Die Stapeldatei ist beendet. 
Die „Eingabeaufforderung“ ist geöffnet und man kann die letzten Meldungen der Stapeldatei sehen.
Als letztes wird die Datei DAUER.TXT im Ordner %ZLW%\%COMPUTERNAME%\ angezeigt:

Start am 30.12.2020 um 18:00 Uhr.

Ende Sicherung C: um 18:20 Uhr.

Ende Sicherung D: um 18:26 Uhr.

pause

Tipps für den Test

Sie brauchen die Stapeldateien nicht abtippen, Sie können diese auf www.eifert.net finden, im Shop
bei der Beschreibung des Buches „Datensicherung für Anfänger“.

Damit ein testdurchlauf nicht zu lange dauert, ändern Sie MAXAGE in Zeile 2 auf einen Wert wie eine
Woche oder einen monat zurück. Die reduzierte Datenmenge sollte zum testen ausreichen. 

Sehr viele Dateien lassen sich nicht lesen, besonders in den Ordnern C:\Program* und
C:\USERS\DEFAULT\ und deshalb dauert die Sicherung sehr lange? Ändern Sie, zumindest für die test-
phase, /R:1 /W:1 in /R:0 (keine wiederholten Leseversuche) und es läuft schneller. Oder tragen Sie
die Pfade in die Ausschlussliste hinter /XD ein. 

Die Sicherung hat ein monatsverzeichnis F:\ASUS-PC\2020-12 angelegt, aber das Verzeichnis ist leer?
Keine Panik. microsoft hat die unterordner versteckt. Warum? Vielleicht damit niemand am Backup herum-
fummelt? machen Sie alle versteckten Ordner sichtbar (siehe Bilder 16.1 und 16.2 im Anhang). Klicken Sie
im Register „Ansicht“ des Windows explorers auf „extras“ → „Optionen“ → „Ordner- und Suchoptionen
ändern“ → Register „Ansicht“ und entfernen Sie den haken vor „geschützte Systemdateien ausblenden
(empfohlen)“. ignorieren Sie die folgende Warnung. Klicken Sie dann auf „übernehmen“ und „OK“. nun
sind die Sicherungsordner sichtbar. 

Anmerkung zum Verwendungszweck

Dies ist eine universelle Stapeldatei, die auf jedem PC ein annehmbares ergebnis bringen sollte. Alle Datei-
typen werden gleich behandelt. Das ist nicht immer optimal. Sie sollten die Stapeldatei für jeden konkreten
PC optimieren. Beispielsweise ist es vermutlich unsinnig, von ihrer Foto- und Filmsammlung jeden monat
eine neue Kopie zu erstellen. um das zu vermeiden, könnten Sie alle Fotos (*.jpg) und Filme (*.mov,
*.avi und andere) von der monatlichen Sicherung ausschließen und sie mit einem separaten Programm
sichern, eventuell in ein Jahresverzeichnis, um Backup-Speicherplatz zu sparen. 

10.6 JAhReSSiCheRung DeR FOtOSAmmLung

Diese Beispieldatei sichert nur Dateien vom typ *.jpg und *.bmp von den beiden Laufwerken QLW1
bis QLW2. Die Sicherung erfolgt nach Laufwerk F: in ein Jahresverzeichnis ohne unterteilung in monate. 

Die ersten Befehle sind wie in 10.3 „Datensicherung (fast ohne) Protokollierung“ 

SET MAXAGE=20180501

rem QLWx = Quelle (ohne Doppelpunkt), ZLW=Ziel (mit Doppelpunkt)

SET QLW1=C

SET QLW2=D

SET ZLW=F:

rem Aus dem Datum wird die Variablen JAHR ausgeschnitten

set JAHR=%date:~-4%

Jetzt beginnt das Erstellen der Ordner und der Kopiervorgang.

echo Bitte einige Minuten Geduld, waehrend die Festplatte durchsucht wird!

TITLE Das Jahresverzeichnis wird synchronisiert

Auf dem Ziellaufwerk (der externen Festplatte) wird eine Ordnerstruktur angelegt: der Netzwerk-
name des PC, gefolgt von Jahreszahl und dem Laufwerk, z. B. F:\ASUS-PC\FOTOS_2020\C

Falls es die Ordner schon gibt, sehen Sie eine Fehlermeldung – ignorieren Sie diese. 
md %ZLW%\%COMPUTERNAME%

md %ZLW%\%COMPUTERNAME%\FOTOS_%JAHR% 

md %ZLW%\%COMPUTERNAME%\FOTOS_%JAHR%\%QLW1%

PAUSE

Jetzt folgt der Kopierbefehl. Das Einsetzen der Werte in die Variablen ergibt
ROBOCOPY C:\ F:\ASUS-PC\FOTOS_2020\C\ *.jpg *.bmp /S /PURGE /FP usw.

ROBOCOPY %QLW1%:\ %ZLW%\%COMPUTERNAME%\FOTOS_%JAHR%\%QLW1%\ *.jpg *.bmp /S

/PURGE /FP /maxage:%MAXAGE% /XA:RSH /XD RECOVER %WINDIR% %ProgramFiles%

RECYCLER $Recycle.Bin "System Volume Information" /w:1 /r:1

PAUSE

rem Pruefen, ob ein zweites Laufwerk gesichert werden soll

if "%QLW2%x" == "x" goto SICHTBAR

Es ist ein weiteres Laufwerk zu sichern. 

TITLE Die zweite Partition %QLW2%: wird gesichert

Ein Ordner wird angelegt, z. B. F:\ASUS-PC\FOTOS_2020\D

md %ZLW%\%COMPUTERNAME%\FOTOS_%JAHR%\%QLW2%

Der zweite Kopierbefehl ist identisch, es wurde nur QLW1 durch QLW2 ersetzt.
ROBOCOPY %QLW2%:\ %ZLW%\%COMPUTERNAME%\FOTOS_%JAHR%\%QLW2%\ *.jpg *.bmp /S

/PURGE /FP /maxage:%MAXAGE% /XA:RSH /XD RECOVER %WINDIR% %ProgramFiles%

RECYCLER $Recycle.Bin "System Volume Information" /w:1 /r:1

PAUSE

:SICHTBAR

@echo off

attrib -h -s %ZLW%\%COMPUTERNAME%\*.* /s /d

Die Variable JAHR werden aus dem Umgebungsspeicher gelöscht.
set JAHR=

:SCHLUSS

pause

Die „Eingabeaufforderung“ ist geöffnet und man kann die letzten Meldungen der Stapeldatei sehen.

137

Anle i tung lokAl



ergänzen Sie die Liste der zu sichernden Dateitypen (im Beispiel sind es nur *.jpg und *.bmp, violett
hervorgehoben) um weitere Dateitypen, die in ihrer Fotosammlung vorkommen. 

Weil die zu sichernden Dateitypen im ROBOCOPY-Befehl ausdrücklich aufgelistet sind, konnte der
Abschnitt /XF (die nicht zu sichernden Dateien) entfallen. 

Zusammenfassung

mit „Jahressicherung“ ist gemeint: Das Programm sichert alle *.jpg und *.bmp Dateien, die auf dem
Laufwerk %QLW1%: und gegebenfalls auf %QLW2%: gespeichert sind. es spielt keine Rolle, aus welchem
Jahr ihre Fotos sind! und wenn Sie auf den Laufwerken %QLW1%: und %QLW1%: neue Ordner anlegen
oder umbenennen oder Fotos löschen, wird der Zielordner bei der nächsten Datensicherung damit in über-
einstimmung gebracht. übrigens befinden sich Bilder auch in den Ordnern Windows und Programme, die
ebenfalls gesichert werden. 

Sobald ein neues Jahr beginnt, wird auf dem Ziellaufwerk ein neuer Jahresordner angelegt. Der Jahresordner
des Vorjahres wird nicht mehr verändert. Alle Bilder des alten Jahresordners sind auch im neuen Jahresordner
enthalten. Wahrscheinlich ist es sinnvoll, den alten Jahresordner auf DVD oder Blu-ray zu brennen und
vom Sicherungsmedium zu löschen. 

Videos sichern

Sie können zur Liste der zu sichernden Dateitypen *.jpg und *.bmp einfach den typ ihrer Videodateien
hinzufügen. Oder Sie sichern die Videodateien separat. Dazu benutzen Sie eine Kopie der obigen Stapeldatei
und ersetzen *.jpg und *.bmp durch *.avi, *.mov, *.mpg, *.mpeg* oder was auch immer. 

Ändern Sie auch den Zielordner. Achten Sie auf die Wahl der Verzeichnisnamen. Wenn der Ordner für
Fotos und der Ordner für Videodateien auf dem gleichen Laufwerk liegen, müssen sie unterschiedliche
namen haben, von denen nicht ein name ein teil des anderen namens ist. Beispielsweise wäre die Paarung
2020-12\Videos und 2020-12\FOTOS OK, doch die Paarung 2020-12 und 2020-12\FOTOS wäre
unzulässig. Warum ist das so?

ROBOCOPY kopiert nicht nur Daten, sondern löscht wegen dem Parameter /PURGE auch Daten aus dem
Backup heraus, die im Quellverzeichnis nicht vorhanden sind. Wenn die Stapeldatei von Kapitel 10.6 einen
Ordner 2020-12\FOTOS benutzen würde, würde diese aus dem Ordner 2020-12 alle unterordner
außer FOTOS herauslöschen. und eine später ausgeführte Batch von Kapitel 10.3 würde zwar alle Daten
des aktuellen monats erneut kopieren, aber dabei auch den unterordner FOTOS herauslöschen. 

gleichfalls ist es unmöglich, Daten von C:\ nach C:\2020\ zu sichern: Die Sicherung würde zyklisch
laufen, solange bis der Datenträger C: voll ist. ein Sichern von C:\USERS\ nach C:\2020\USERS\ ist
problemlos möglich (aber ist es sinnvoll, Original und Backup auf der gleichen Festplatte zu haben?). 

